On the Formalization of Fourier Transform in Higher-order Logic (Rough Diamond)

Adnan Rashid and Osman Hasan

School of Electrical Engineering and Computer Science (SEECS)
National University of Sciences and Technology (NUST)
Islamabad, Pakistan
{adnan.rashid,osman.hasan}@seecs.nust.edu.pk

Abstract. Fourier transform based techniques are widely used for solving differential equations and to perform the frequency response analysis of signals in many safety-critical systems. To perform the formal analysis of these systems, we present a formalization of Fourier transform using higher-order logic. In particular, we use the HOL-Light's differential, integral, transcendental and topological theories of multivariable calculus to formally define Fourier transform and reason about the correctness of its classical properties, such as existence, linearity, frequency shifting, modulation, time reversal and differentiation in time-domain. In order to demonstrate the practical effectiveness of the proposed formalization, we use it to formally verify the frequency response of an automobile suspension system.

Keywords: Higher-order Logic, HOL-Light, Fourier Transform.

1 Introduction

It is customary to use differential equations for capturing the dynamic behavior of engineering and physical systems for their continuous-time analysis [9]. The complexity of the analysis varies with their size, nature of the input signals and the design constraints. Fourier Transform [2] is a transform method, which converts a time varying function to its corresponding ω-domain representation, where ω is its corresponding angular frequency [1]. In this way, the differentiation and integration in time domain analysis are transformed into multiplication and division operators in the frequency domain and thus are easily solved through algebraic manipulation. Moreover, the ω-domain representations of the differential equations can also be used for the frequency response analysis of the corresponding systems.

The first step in the continuous-time system analysis, using Fourier transform, is to model the dynamics of the system using a differential equation. This differential equation is then transformed into its equivalent ω-domain representation by using the Fourier transform. Next, the resulting ω-domain equation is simplified using various Fourier transform properties, such as existence, linearity, frequency shifting, modulation, time reversal and differentiation. The main purpose is to either solve the differential equation to obtain values for the variable ω or obtain the frequency response of the system corresponding to the given differential equation. Once the frequency response is obtained, it can be used to analyze the dynamics of the system by studying the impact of different frequency components on the intended behaviour of the given system.

Traditionally, the transform methods based analysis has been done using paper-and-pencil, numerical methods and symbolic techniques. However, all of these techniques cannot ascertain accurate analysis due to their inherent limitations, like human-error proneness, numerical errors and discretization errors. Given the wide-spread usage of physical systems in many safety-critical domains, such as medicine and transportation, accurate transform methods based analysis has become a dire need. With the same motivation, higher-order-logic theorem proving has been used for the formalization of Z [7] and Laplace [8] transforms. However, the formalization of Z-transform can only be utilized for discrete-time system analysis. Similarly, Laplace transform based analysis is only limited to causal functions, i.e., the functions that fulfill the condition: $f(x) = 0$ for all $x < 0$. Physical systems are often modeled by the non-causal continuous functions, i.e., the functions with infinite extent. Fourier transform can cater for the analysis involving both continuous and non-causal functions and thus can overcome the above-mentioned limitations of Z and Laplace transforms.

In this paper, we propose to formalize Fourier transform in higher-order logic to leverage upon its benefits for formally analyzing physical continuous-time linear systems. In particular, we formalize the definition of Fourier transform in higher-order logic and use it to verify the classical properties of Fourier transform, such as existence, linearity, frequency shifting, modulation, time reversal and differentiation. These foundations can be built upon to reason about the analytical solutions of differential equations or frequency responses of the physical systems. In order to demonstrate the practical effectiveness of the reported formalization, we present a formal analysis of an automobile suspension system.

2 Formalization of Fourier Transform

Mathematically, the Fourier transform is defined for a function $f : \mathbb{R}^1 \rightarrow \mathbb{C}$ as:

$$\mathcal{F}[f(t)] = F(\omega) = \int_{-\infty}^{+\infty} f(t) e^{-j\omega t} dt, \ \omega \ \epsilon \ \mathbb{R} \tag{1}$$

We formalize Equation 1 in HOL-Light as follows:

Definition 1. Fourier Transform

```
⊢ ∀ w f. fourier f w =
    integral UNIV (λt. cexp (--((ii * Cx w) * Cx (drop t))) * f t)
```

The function `fourier` accepts a complex-valued function $\mathtt{f} : \mathbb{R}^1 \rightarrow \mathbb{R}^2$ and a real number `w` and returns a complex number that is the Fourier transform of `f` as represented by Equation 1. In the above function, we used complex exponential function `cexp` : $\mathbb{R}^2 \rightarrow \mathbb{R}^2$ because the return data-type of the function `f` is $\mathbb{R}^2$. To multiply `w` with `ii`, we first converted `w` into a complex number $\mathbb{R}^2$ using `Cx`. Similarly, `t` has data-type $\mathbb{R}^1$ and to multiply it with `ii * Cx w`, it is first converted into a real number by using `drop` and then it is converted to data-type $\mathbb{R}^2$ using `Cx`. Next, we use the vector function `integral` to integrate the expression $f(t)e^{-i\omega t}$ over the whole real line since the data-type of this expression is $\mathbb{R}^2$. Since the region of integration of the vector integral function must be a vector space, therefore we represented the interval of the integral by `UNIV` : $\mathbb{R}^1$ which represents the whole real line.

The Fourier transform of a function f exists, i.e., the integrand of Equation 1 is integrable, and the integral has some converging limit value, if f is piecewise smooth and is absolutely integrable on the whole real line [1,5]. A function is said to be piecewise smooth on an interval if it is piecewise differentiable on that interval. Similarly, a function f is absolutely integrable on the whole real line if it is absolutely integrable on both the positive and negative real lines. The Fourier existence condition can thus be formalized in HOL-Light as follows:

Definition 2. Fourier Exists
```
⊢ ∀ f g w a b. fourier_exists f =
   (∀ a b. f piecewise_differentiable_on interval [lift a, lift b]) ∧
           f absolutely_integrable_on {x | &0 <= drop x} ∧
           f absolutely_integrable_on {x | drop x <= &0}
```

In the above function, the first conjunct expresses the piecewise smoothness condition for the function `f`. In the second conjunct, `{x | &0 <= drop x}` represents the interval $[0, \infty)$, whereas `{x | drop x <= &0}` represents the interval $(-\infty, 0]$ in the last conjunct. Both these conjuncts jointly ensure that the function f is absolutely integrable on whole real line.

3 Formal Verification of Fourier Transform Properties

In this section, we use Definitions 1 and 2 to verify some of the classical properties of Fourier transform in HOL-Light. The verification of these properties not only ensures the correctness of our definitions but also plays a vital role in minimizing the user intervention and time consumption in reasoning about Fourier transform based analysis of systems.

The existence of the improper integral of Fourier Transform is a pre-condition for most of the arithmetic manipulations involving the Fourier transforms. This condition is formalized in HOL-Light as follows:

Theorem 1. Integrability of Integrand of Fourier Transform Integral
```
⊢ ∀ f w. fourier_exists f ⇒
  (λt. cexp (--((ii * Cx w) * Cx (drop t))) * f t) integrable_on UNIV
```

The proof of above theorem is based on splitting of the region of integration, i.e., the whole real line `UNIV` : $\mathbb{R}^1$, as a union of positive real line (interval

$[0, \infty)$) and negative real line (interval $(-\infty, 0]$). Then, some theorems regarding integration and integrability are used to conclude the proof of Theorem 1.

Next, we verified some of the classical properties of Fourier transform, given in Table 1.

Table 1: Properties of Fourier Transform

Mathematical Form	Formalized Form
Linearity	
$\mathcal{F}[\alpha f(t) + \beta g(t)] = \alpha F(\omega) + \beta G(\omega)$	⊢ ∀ f g w a b. fourier_exists f ∧ fourier_exists g ⇒ fourier (λt. a ∗ f t + b ∗ g t) w = a ∗ fourier f w + b ∗ fourier g w
Frequency Shifting	
$\mathcal{F}[e^{i\omega_0 t} f(t)] = F(\omega - \omega_0)$	⊢ ∀ f w w0. fourier_exists f ⇒ fourier (λt. cexp ((ii ∗ Cx (w0)) ∗ Cx (drop t)) ∗ f t) w = fourier f (w - w0)
Modulation	
$\mathcal{F}[cos(\omega_0 t) f(t)] = \dfrac{F(\omega - \omega_0) + F(\omega + \omega_0)}{2}$	⊢ ∀ f w w0. fourier_exists f ⇒ fourier (λt. ccos (Cx w0 ∗ Cx (drop t)) ∗ f t) w = (fourier f (w - w0) + fourier f (w + w0)) / Cx (&2)
$\mathcal{F}[sin(\omega_0 t) f(t)] = \dfrac{F(\omega - \omega_0) - F(\omega + \omega_0)}{2i}$	⊢ ∀ f w w0. fourier_exists f ⇒ fourier (λt. csin (Cx w0 ∗ Cx (drop t)) ∗ f t) w = (fourier f (w - w0) - fourier f (w + w0)) / (Cx (&2) ∗ ii)
Time Reversal	
$\mathcal{F}[f(-t)] = F(-\omega)$	⊢ ∀ f w. fourier_exists f ⇒ fourier (λt. f (--t)) w = fourier f (--w)
First-order Differentiation	
$\mathcal{F}[\frac{d}{dt} f(t)] = i\omega F(\omega)$	⊢ ∀ f w. fourier_exists f ∧ fourier_exists (λt. vector_derivative f (at t)) ∧ (∀t. f differentiable at t) ∧ ((λt. f (lift t)) → vec 0) at_posinfinity ∧ ((λt. f (lift t)) → vec 0) at_neginfinity ⇒ fourier (λt. vector_derivative f (at t)) w = ii ∗ Cx w ∗ fourier f w
Higher-order Differentiation	
$\mathcal{F}[\frac{d^n}{dt^n} f(t)] = (i\omega)^n F(\omega)$	⊢ ∀ f w n. fourier_exists_higher_deriv n f ∧ (∀t. differentiable_higher_derivative n f t) ∧ (∀p. p < n ⇒ ((λt. higher_vector_derivative p f (lift t)) → vec 0) at_posinfinity) ∧ (∀p. p < n ⇒ ((λt. higher_vector_derivative p f (lift t)) → vec 0) at_neginfinity) ⇒ fourier (λt. higher_vector_derivative n f t) w = (ii ∗ Cx w) pow n ∗ fourier f w

The above-mentioned formalization is done interactively and it took around 4000 lines of code and approximately 600 man-hours. The first author started working with HOL-Light as a novice user and it took him about 200 man-hours to get familiar with its proof styles and procedures. About another 100 man-hours were spent in understanding the Multivariate theories of HOL-Light, which are the foundational theories towards this work. The actual formalization task took about 300 man-hours. The major difficulty faced during the formalization was the unavailability of detailed proofs for the properties of Fourier transform in literature. The available paper-and-pencil based proofs were found to be very abstract and missing the complete reasoning about the steps. The source code of our formalization is available for download [6] and can be utilized for further developments and the analysis of physical systems.

4 Application: Automobile Suspension System

In this section, we provide the verification of the frequency response of an automobile suspension system, depicted in Figure 1. An automobile suspension system consists of the chassis connected to the wheels through a spring and dashpot (shock absorber). The road surface can be thought of as a superposition of rapid and gradual small-amplitude changes in elevation, which represents the roughness of the surface. These rapid and gradual changes are acting like high and low frequencies, respectively. The automobile suspension system is intended to filter out the rapid variations on the road surface, i.e., to act as a low pass filter. We perform the formal analysis of this system using our proposed formalization of Fourier transform within the sound core of HOL-Light theorem prover.



Fig. 1: Automobile Suspension System [5]

The behaviour of a automobile suspension system with input $u(t)$ and output $y(t)$ can be expressed by the following differential equation [5]:

$$M\frac{d^2y(t)}{dt^2} + b\frac{dy(t)}{dt} + ky(t) = ku(t) + b\frac{du(t)}{dt}, \quad (2)$$

In the above equation, M is the mass of the chassis, whereas, k is the spring constant and b represents the shock absorber constant, as shown in Figure 1. All of these are design parameters of the underlying system and can have positive values only.

The corresponding frequency response of the automobile suspension system is given as follows [5]:

$$\frac{Y(\omega)}{U(\omega)} = \frac{\frac{b}{M}(i\omega) + \frac{k}{M}}{(i\omega)^2 + \frac{b}{M}(i\omega) + \frac{k}{M}} \tag{3}$$

We aim to verify this frequency response using Equation 2, which can be verified as the following theorem in HOL-Light.

Theorem 2. Frequency Response of Automobile Suspension System

```
⊢ ∀ y u w a. &0 < M ∧ &0 < b ∧ &0 < k ∧
  (∀t. differentiable_higher_derivative 2 y t) ∧
  (∀t. differentiable_higher_derivative 1 u t) ∧
  fourier_exists_higher_deriv 2 y ∧
  fourier_exists_higher_deriv 1 u ∧
  (∀p. p < 2 ⇒
       ((λt. higher_vector_derivative p y (lift t)) → vec 0)
         at_posinfinity) ∧
  (∀p. p < 2 ⇒
       ((λt. higher_vector_derivative p y (lift t)) → vec 0)
         at_neginfinity) ∧
  ((λt. u (lift t)) → vec 0) at_posinfinity ∧
  ((λt. u (lift t)) → vec 0) at_neginfinity ∧
  (∀t. diff_eq_ASS y u a b c) ∧ ∼(fourier u w = Cx (&0)) ∧
  ∼((ii ∗ Cx w) pow 2 + Cx (b / M) ∗ ii ∗ Cx w
                                    + Cx (k / M) = Cx (&0))
   ⇒ (fourier y w / fourier u w =
                       (Cx (b / M) ∗ ii ∗ Cx w + Cx (k / M)) /
     ((ii ∗ Cx w) pow 2 + Cx (b / M) ∗ ii ∗ Cx w + Cx (k / M))
```

The first three assumptions ensure that the variables corresponding to mass of chassis (M), spring constant (k) and shock absorber constant (b) cannot be negative or zero. The next two assumptions ensure that the functions y and u are differentiable up to the second-order and first-order, respectively. The next assumption represents the Fourier transform existence condition upto the second-order derivatives of function y. Similarly, the next assumption ensures that the Fourier transform exists up to the first-order derivative of function u. The next two assumptions represent the condition $\lim_{t\to\pm\infty} y^{(k)}(t) = 0$ for each $k = 0, 1$, i.e., $\lim_{t\to\pm\infty} y^{(1)}(t) = 0$ and $\lim_{t\to\pm\infty} y^{(0)}(t) = \lim_{t\to\pm\infty} y(t) = 0$, where $y^{(k)}$ is the k^{th} derivative of y. The next two assumptions provide the condition $\lim_{t\to\pm\infty} u(t) = 0$. The next assumption represents the formalization of Equation 2 and the last two assumptions provide some interesting design related relationships, which must hold for constructing a reliable automobile suspension system. Finally, the conclusion of the above theorem represents the frequency response given by Equation 3. The proof of Theorem 2 is based on Definition 1 and the property of Fourier transform of higher-order derivative of a function, along with some arithmetic reasoning. The proof script for this application consists of approximately 500 lines of HOL-Light code [6] and the proof process took just a couple of hours, which clearly indicates the usefulness of our proposed formalization in conducting the Fourier transform analysis of real-world applications. Given the continuous and non-causal nature of the functions involved in this analysis, the existing Z [7] and Laplace transform [8] formalizations cannot be used for conducting the above-mentioned formal analysis.

5 Conclusions

In this paper, we proposed a formalization of Fourier transform in higher-order logic. We presented the formal definition of Fourier transform and based on it, verified its properties, namely existence, linearity, frequency shifting, modulation, time reversal and differentiation in time-domain. Lastly, in order to demonstrate the practical effectiveness of the proposed formalization, we presented a formal analysis of an automobile suspension system.

The proposed formalization of Fourier transform can be utilized to conduct the formal analysis of many safety-critical systems involving signal processing filters, such as low-pass, high-pass, band-pass and band-stop [5] and in wireless communication systems, such as antenna [2] and signal transmission [3]. Similarly, in optics, it can be used to formally study the behaviour of light, such as intensity and diffraction, in different optical devices [2], which can be very useful for the recently initiated project on the usage of higher-order-logic theorem proving for the formal analysis of optics [4].

References

1. Beerends, R.J., Morsche, H.G., Van den Berg, J.C., Van de Vrie, E.M.: Fourier and Laplace Transforms. Cambridge University Press, Cambridge (2003)
2. Bracewell, R.N.: The Fourier Transform and its Applications. McGraw-Hill New York (1978)
3. Du, K.L., Swamy, M.N.S.: Wireless Communication Systems: from RF Subsystems to 4G Enabling Technologies. Cambridge University Press, Cambridge, U.K. (2010)
4. Khan-Afshar, S., Siddique, U., Mahmoud, M.Y., Aravantinos, V., Seddiki, O., Hasan, O., Tahar, S.: Formal Analysis of Optical Systems. Mathematics in Computer Science 8(1), 39–70 (2014)
5. Oppenheim, A.V., Willsky, A.S., Hamid Nawab, S.: Signals and Systems. Prentice Hall Processing series, Prentice Hall, Inc. U.S.A, 2 edn. (1996)
6. Rashid, A.: On the Formalization of Fourier Transform in Higher-order Logic. `http://save.seecs.nust.edu.pk/projects/fourier/` (2016)
7. Siddique, U., Mahmoud, M.Y., Tahar, S.: On the Formalization of Z-Transform in HOL. In: Interactive Theorem Proving, pp. 483–498. Springer (2014)
8. Taqdees, S.H., Hasan, O.: Formalization of Laplace Transform Using the Multivariable Calculus Theory of HOL-Light. In: Logic for Programming, Artificial Intelligence, and Reasoning. pp. 744–758. Springer (2013)
9. Yang, X.S.: Mathematical Modeling with Multidisciplinary Applications. John Wiley and Sons (2013)